
Wetterdaten aufzeichnen Seite 1
hartmut-waller.info

Die mit dem Temperatursensor DHT11/DHT22 gemessene Temperatur soll mit aktuellem Datum und 
aktueller Zeit auf eine SD-Karte gespeichert werden.

So soll es aussehen:

Benötigte Bauteile:

RTC-Modul DS3231
Temperatur-/Feuchtigkeitssensor DHT11/DHT22
SD-Kartenleser (Datenlogger-Shield oder Ethernet-Shield)
Leitungsdrähte

Beispiele für Temperatursensoren

Die Pinbelegung kann sich von der hier gezeigten unterscheiden. Achte 
auf die Beschriftung auf dem Modul!

Achte auf die Pin-Belegung des SD-Kartenlesers!
Die SD-Karte muss mit FAT32 formatiert sein!
SDO = Serial Data Out
SDI = Serial Data In

Baue die Schaltung auf.

Benötigte Bibliotheken:

Sketch → Bibliothek einbinden → Bibliotheken verwalten

CS → 4
SCK → 13
SDO → 11
SDI → 12
VCC → 5V
GND → GND

CS → 4
SCK → 13
SDO → 11
SDI → 12
VCC → 5V
GND → GND

Im Kopf des Programms werden die benötigten Bibliotheken eingebunden und die Variablen definiert.
Beachte die Kommentare:

#include "SdFat.h"

#include "DHT.h"

// Bezeichnung der SD-Karte

SdFat SD;

// Pin des DHT-Sensors

int SENSOR_DHT = 8;

// Sensortyp festlegen

// DHT22 oder DHT11

#define SensorTyp DHT11

// Sensor DHT einen Namen zuweisen

DHT dht(SENSOR_DHT, SensorTyp);

// Bibliothek für das RTC-Modul

# include <RTClib.h>

RTC_DS3231 rtc;

// Bezeichnung der Textdatei

File Temperaturmessung;

// Datenpin für das SD-Kartenmodul

// int DatenPin = 4;

// Datenpin für das Datenlogger-Shield

int DatenPin = 10;

Der setup-Teil initialisiert die SD-Karte, startet das RTC-Modul  und legt den pinMode für den Sensor 
fest.

void setup()

{

  Serial.begin(9600);

  // auf serielle Verbindung warten

  while (!Serial) {;}

  Serial.println(F("Initialisiere SD-Karte"));

  if (!SD.begin(DatenPin))

  {

    Serial.println(F("Initialisierung fehlgeschlagen!"));

  }

  else Serial.println(F("Initialisierung abgeschlossen"));

  // rtc starten

  rtc.begin();

  /*

    wenn Datum und Zeit nicht korrekt → Datum/Zeit setzen

    Jahr, Monat, Tag, Stunde, Minute, Sekunde

    rtc.adjust(DateTime(2023, 5, 15, 10, 30, 30));

  */

  // Sensor DHT starten

  dht.begin();

}

Der loop-Teil: Beachte die Kommentare.

void loop()

{

  DateTime aktuell = rtc.now();

 

  // Daten lesen

  float Temperatur = dht.readTemperature();

  float Luftfeuchtigkeit = dht.readHumidity();

  // Funktion aufrufen

  ZeitAusgeben(aktuell);

  // in Strings umwandeln, . durch , ersetzen

  String AnzeigeTemperatur = String(Temperatur);

  AnzeigeTemperatur.replace(".", ",");

  String AnzeigeLuftfeuchtigkeit = String(Luftfeuchtigkeit);

  AnzeigeLuftfeuchtigkeit.replace(".", ",");

  Serial.print(F("Temperatur:\t\t"));

  Serial.println(AnzeigeTemperatur);

  Serial.print(F("Luftfeuchtigkeit in %:\t"));

  Serial.println(AnzeigeLuftfeuchtigkeit);

  

  /*

    in Datei schreiben

    O_CREAT -> Datei erstellen, wenn sie nicht existiert

    O_WRITE -> in die Datei schreiben

    O_AT_END -> Startposition zum Schreiben an das Ende der Datei setzen

  */

  if (!Temperaturmessung.open("Messung.csv", O_CREAT | O_WRITE | O_AT_END))

  {

    Serial.print(F("Datei kann nicht ge\u00f6ffnet werden!"));

  }

  // wenn die Datei geöffnet werden konnte ...

  if (Temperaturmessung)

  {

    Serial.println(F("Schreibe Messdaten in die Datei Messung.csv ... "));

    Serial.println(F("-----------------------------------"));

    // Überschrift in Datei schreiben

    // Funktion aufrufen

    schreibeUeberschrift();

    Temperaturmessung.print(AktuellesDatum + ";");

    Temperaturmessung.print(AktuelleZeit + ";");

    Temperaturmessung.print(AnzeigeTemperatur + ";");

    Temperaturmessung.print(AnzeigeLuftfeuchtigkeit);

    Temperaturmessung.println();

    // Schreibfehler abfragen

    if (!Temperaturmessung.sync() || Temperaturmessung.getWriteError())

    {

      Serial.print(F("Schreibfehler!"));

    }

    else

    {

      // Datei schließen

      Temperaturmessung.close();

      Serial.println(F("Abgeschlossen."));

      Serial.println();

    }

  }

  // Zeit bis zur nächsten Messung (eine Minute)

  delay(60000);

}

Im loop-Teil werden zwei Methoden aufgerufen:

void ZeitAusgeben(DateTime aktuell)

{

  char Datum[] = "DD.MM.YYYY";

  AktuellesDatum = aktuell.toString(Datum);

  char Zeit[] = "Uhrzeit: hh:mm:ss";

  AktuelleZeit = aktuell.toString(Zeit);

  Serial.println(AktuellesDatum + " " + AktuelleZeit);

}

void schreibeUeberschrift()

{

  Temperaturmessung.println();

  Temperaturmessung.print(F("Datum"));

  // ; als Trennzeichen für die CSV-Datei

  Temperaturmessung.print(";");

  Temperaturmessung.print(F("Zeit"));

  Temperaturmessung.print(";");

  Temperaturmessung.print(F("Temperatur in °C"));

  Temperaturmessung.print(";");

  Temperaturmessung.print(F("Luftfeuchtigkeit in %"));

  Temperaturmessung.println();

}

